Spontaneous Parametric Down Conversion Calculations

Enrique J. Galvez 12/31/17

Constants

In[61]:=
```
lamp = 0.40236 (* pump wavelength in um *)
```
Out[61]=
```
0.40236
```

In[62]:=
```
lamddc = 2 * lamp (* wavelength of degenerate SPDC photons *)
```
Out[62]=
```
0.80472
```

In[63]:=
```
thetapmbbo =
 29.425 * Degree (* BBO phase matching angle for SPDC at 3 degree in rad *)
```
Out[63]=
```
0.513563
```

In[64]:=
```
thetapmbbocoll = 29.005497 * Degree
```
Out[64]=
```
0.506241
```

Index of refraction of BBO

In[65]:=
```
no[lum_] := Sqrt[2.7359 + 0.01878 / (lum^2 - 0.01822) - 0.01354 * lum^2]
 (* ordinary index*)
```

In[66]:=
```
ne[lum_] := Sqrt[2.3753 + 0.01224 / (lum^2 - 0.01667) - 0.01516 * lum^2]
 (* extraordinary index *)
```

In[67]:=
```
npe[theta_, lum_] := 1 / Sqrt[Cos[theta]^2 / no[lum]^2 + Sin[theta]^2 / ne[lum]^2]
  (* extraordinary index of tilted crystal *)
```

In[68]:=
```
no[lamddc]
```
Out[68]=
```
1.66041
```

In[69]:=
```
npe[thetapmbbocoll, lamp] (* verifying phase matching *)
```
Out[69]=
```
1.66041
```

In[70]:=
```
Plot[{no[lum / 1000], ne[lum / 1000], npe[thetapmbbo, lum / 1000]},
 {lum, 350, 900}, AxesLabel → {"λ (nm)", "n"}, PlotRange → {1.5, 1.75},
 PlotLabels → {"ordinary", "extraordinary", "phase-matched"},
 GridLines -> Automatic]
```

Out[70]=



Indices of refraction of BBO. The pump-beam is phase-matched such that its index of refraction (at 405 nm) matches the index of refraction of the down-converted photons at 810 nm.

Photon Correlations

In[71]:=
```
lami[lams_] := 1 / (1 / lamp - 1 / lams)
  (* idler wavelength in terms of signal, from energy conservation *)
```

In[72]:=
```
lami[lamddc](* just checking *)
```

Out[72]=
```
0.80472
```

In[73]:=
```
Plot[lami[lamsnm / 1000] * 1000, {lamsnm, 790, 820},
 GridLines -> Automatic, AxesLabel → {"λs", "λi"}]
```

Out[73]=



Graph of the wavelength of the idler as a function of the wavelength of the signal (in nm).

In[74]:=
```
kp[thetapm_] := 2 * Pi * npe[thetapm, lamp] / lamp (* momentum of pump *)
```

In[75]:=
```
kp[thetapmbbo]
```
Out[75]=
```
25.9158
```

In[76]:=
```
ks[lams_] := 2 * Pi * no[lams] / lams (* momentum of signal *)
```

In[77]:=
```
ks[lamddc]
```
Out[77]=
```
12.9644
```

In[78]:=
```
ki[lams_] := 2 * Pi * no[lami[lams]] / lami[lams] (* momentum of the idler *)
```

In[79]:=
```
ki[lamddc]
```
Out[79]=
```
12.9644
```

In[80]:=
```
kp[thetapmbbocoll] / 2 / ki[lamddc]
```
Out[80]=
```
1.
```

In[81]:=
```
kp[thetapmbbo] / 2 / ki[lamddc]
```
Out[81]=
```
0.999503
```

In[82]:=
```
thetai[thetapm_, lams_] :=
 ArcCos[(kp[thetapm]^2 - ks[lams]^2 + ki[lams]^2) / (2 * kp[thetapm] * ki[lams])]
 (*angle formed by the idler photon, per conservation of momentum inside xtal *)
```

In[83]:=
```
thetas[thetapm_, lams_] :=
 ArcCos[(kp[thetapm]^2 - ki[lams]^2 + ks[lams]^2) / (2 * kp[thetapm] * ks[lams])]
```

In[84]:=
```
thetai[thetapmbbo, 2 * lamp] / Degree // N
```
Out[84]=
```
1.80693
```

In[85]:=
```
thetaiout[thetapm_, lams_] := ArcSin[no[lami[lams]] * Sin[thetai[thetapm, lams]]]
 (* angle outside of the crystal *)
```

In[86]:=
```
thetaiout[thetapmbbo, 2 * lamp] / Degree
```
Out[86]=
```
3.00112
```

In[87]:=
```
thetasout[thetapm_, lams_] := ArcSin[no[lami[lams]] * Sin[thetas[thetapm, lams]]]
 (* angle that the signal makes as a function of signal wavelength *)
```

In[88]:=
```
Plot[{thetaiout[thetapmbbo, (lamnm / 1000)] / Degree,
  thetasout[thetapmbbo, (lamnm / 1000)] / Degree}, {lamnm, 750, 890},
 GridLines → Automatic, AxesLabel → {"λs", "θ"} , PlotLabels → {"idler", "signal"}]
```

Out[88]=



SPDC angle of idler and signal (in degrees) as a function of the wavelength of signal (in nm). The degenerate wavelength is λ_{dc} (lamdc). The two curves cross at $\lambda_s = \lambda_i = \lambda_{dc}$. When $\lambda_s < \lambda_{dc}$ then $\lambda_i > \lambda_{dc}$. Therefore $k_s > k_i$, so $\theta_s < \theta_i$ from conservation of momentum in the y direction. This extends to outside the crystal but at larger angles due to refraction from the crystal to air. A caustic appears due to the variation of the index of refraction with wavelength.

In[89]:=
```
ks[lamddc]
```

Out[89]=
```
12.9644
```

In[90]:=
```
ki[lamddc]
```

Out[90]=
```
12.9644
```

In[91]:=
```
lamddcm = 780 / 1000 // N
```

Out[91]=
```
0.78
```

In[92]:=
```
ki[lamddcm]
```

Out[92]=
```
12.5478
```

In[93]:=
```
ks[lamddcm]
```

Out[93]=
```
13.3813
```

In[94]:=
```
thetai[thetapmbbo, lamddcm] / Degree
```

Out[94]=
```
1.89526
```

In[95]:=
```
thetas[thetapmbbo, lamddcm] / Degree
```

Out[95]=
```
1.77717
```

In[96]:=
```
ki[lamddcm] * Cos[thetai[thetapmbbo, lamddcm]]
```

Out[96]=
```
12.5409
```

In[97]:=
```
ks[lamddcm] * Cos[thetas[thetapmbbo, lamddcm]]
```
Out[97]= 13.3749

In[98]:=
```
ki[lamddcm] * Cos[thetai[thetapmbbo, lamddcm]] +
 ks[lamddcm] * Cos[thetas[thetapmbbo, lamddcm]]
```
Out[98]= 25.9158

In[99]:=
```
25.915836937837334
```
Out[99]= 25.9158

In[100]:=
```
kp[thetapmbbo]
```
Out[100]= 25.9158

In[101]:=
```
ki[lamddcm] * Sin[thetai[thetapmbbo, lamddcm]]
```
Out[101]= 0.414988

In[102]:=
```
ks[lamddcm] * Sin[thetas[thetapmbbo, lamddcm]]
```
Out[102]= 0.414988

In[103]:=
```
Plot[{thetaiout[thetapmbbocoll, (lamnm / 1000)] / Degree,
  thetaiout[thetapmbbo, (lamnm / 1000)] / Degree,
  thetaiout[thetapmbbo + 0.5 * Degree, (lamnm / 1000)] / Degree},
 {lamnm, 790, 820}, GridLines → Automatic, AxesLabel → {"λs", "θ"} ,
 PlotLabels → {"collinear", "3 degrees", "4.2 degrees"}]
```
Out[103]=



In[104]:=
```
Plot[thetaiout[thetapmbbo, (lamnm / 1000)] * 1000,
 {lamnm, 795, 820}, GridLines → Automatic , AxesLabel → {"λs", "θiL"}]
```

Out[104]=



Position of SPDC along an arc 1-m in radius from crystal

Tuning

In[111]:=
```
Plot[thetaiout[thetangle / 180 * Pi, lamddc] / Degree, {thetangle, 29, 29.5},
 GridLines → Automatic , AxesLabel → {"θphase-match", "θout"}]
```

Out[111]=